



Techniques of Water-Resources Investigations  
of the United States Geological Survey

Chapter A1

**A MODULAR THREE-DIMENSIONAL  
FINITE-DIFFERENCE GROUND-WATER  
FLOW MODEL**

By Michael G. McDonald and  
Arlen W. Harbaugh

This chapter supersedes U.S. Geological  
Survey Open-File Report 83-875

Book 6

MODELING TECHNIQUES

## Narrative for Module BAS10T

Module BAS10T invokes submodules which write results of the simulation. Those results include head, drawdown, overall volumetric budget, and a time summary. Results are printed according to flags IHDDFL, IOFLG, and IBUDFL which are set by module BAS10C (Output Control). If flag IHDDFL is set, a table of flags named IOFLG is used to determine which heads and drawdown should be written (printer or disk) and for which layers it should be written. This module (BAS10T) calls submodules SBAS1H and SBAS1D to write heads and drawdowns respectively. If flag IBUDFL is set, submodule SBAS1V is invoked to calculate and print the overall volumetric budget. After every time step during which results have been printed, a time summary is printed.

Module BAS10T performs its functions in the following order:

1. Clear flag IPFLG. This flag is set later in this module if any results are printed. It controls the printing of a time summary.
2. If the iterative procedure failed to converge, print a message to that effect.
3. If the head and drawdown flag (IHDDFL) are set, call submodules SBAS1H and SBAS1D to write heads and drawdowns in accordance with the flags in the table IOFLG.
4. If the budget flag (IBUDFL) is set, call submodule SBAS1V to calculate and print the volumetric budget.
5. If the printout flag (IPFLG) is set, call submodule SBAS1T to print a time summary.
6. RETURN.

## Flow Chart for Module BAS10T

IPFLG is the printout flag. It is set when any results are printed. If it is set, a time summary is printed.

SBAS1H is a submodule which writes heads.

SBAS1D is a submodule which writes drawdown.

SBAS1V is a submodule which prints the volumetric budget.


SBAS1T is a submodule which prints a time summary.

IHDDFL is the head/drawdown flag. If it is set, heads and drawdown will be written in accordance with flag settings in IOFLG.

IBUDFL is the budget print flag. If it is set, volumetric budget will be printed.

IOFLG is a table of flags with one entry for each layer. Each entry has four flags:

- 1---head print
- 2---drawdown print
- 3---head save
- 4---drawdown save



```

SUBROUTINE BAS1OT(HNEW, STRT, ISTRT, BUFF, IOFLG, MSUM, IBOUND, VBNM,
1  VBVL, KSTP, KPER, DELT, PERTIM, TOTIM, ITMUNI, NCOL, NROW, NLAY, ICNVG,
2  IHDDFL, IBUDFL, IHEDFM, IHEDUN, IDDNFM, IDDNUN, IOUT)
C-----VERSION 1522 12MAY1987 BAS1OT
C *****
C OUTPUT TIME, VOLUMETRIC BUDGET, HEAD, AND DRAWDOWN
C *****
C
C SPECIFICATIONS:
C -----
C CHARACTER*4 VBNM
C DOUBLE PRECISION HNEW
C
C DIMENSION HNEW(NCOL, NROW, NLAY), STRT(NCOL, NROW, NLAY),
1          VBNM(4,20), VBVL(4,20), IOFLG(NLAY,4),
2          IBOUND(NCOL, NROW, NLAY), BUFF(NCOL, NROW, NLAY)
C -----
C
C1-----CLEAR PRINTOUT FLAG (IPFLG)
C          IPFLG=0
C
C2-----IF ITERATIVE PROCEDURE FAILED TO CONVERGE PRINT MESSAGE
C          IF(ICNVG.EQ.0) WRITE(IOUT,1) KSTP,KPER
C          1 FORMAT(1H0,10X,'****FAILED TO CONVERGE IN TIME STEP',I3,
C          1          ' OF STRESS PERIOD',I3,'****')
C
C3-----IF HEAD AND DRAWDOWN FLAG (IHDDFL) IS SET WRITE HEAD AND
C3-----DRAWDOWN IN ACCORDANCE WITH FLAGS IN IOFLG.
C          IF(IHDDFL.EQ.0) GO TO 100
C
C          CALL SBAS1H(HNEW, BUFF, IOFLG, KSTP, KPER, NCOL, NROW,
C          1          NLAY, IOUT, IHEDFM, IHEDUN, IPFLG, PERTIM, TOTIM)
C          CALL SBAS1D(HNEW, BUFF, IOFLG, KSTP, KPER, NCOL, NROW, NLAY, IOUT,
C          1          IDDNFM, IDDNUN, STRT, ISTRT, IBOUND, IPFLG, PERTIM, TOTIM)
C
C4-----PRINT TOTAL BUDGET IF REQUESTED
C          100 IF(IBUDFL.EQ.0) GO TO 120
C          CALL SBAS1V(MSUM, VBNM, VBVL, KSTP, KPER, IOUT)
C          IPFLG=1
C
C5-----END PRINTOUT WITH TIME SUMMARY AND FORM FEED IF ANY PRINTOUT
C5-----WILL BE PRODUCED.
C          120 IF(IPFLG.EQ.0) RETURN
C          CALL SBAS1T(KSTP, KPER, DELT, PERTIM, TOTIM, ITMUNI, IOUT)
C          WRITE(IOUT,101)
C          101 FORMAT(1H1)
C
C6-----RETURN
C          RETURN
C          END

```

List of Variables for Module BAS10T

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
BUFF	Global	DIMENSION (NCOL,NROW,NLAY), Buffer used to accumulate information before printing or recording it.
DELT	Global	Length of the current time step.
HNEW	Global	DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration.
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IBUDFL	Package	Flag. = 0, volumetric budget will not be printed for the current time step. ≠ 0, volumetric budget should be printed for the current time step.
ICNVG	Global	Flag is set equal to one when the iteration procedure has converged.
IDDNFM	Package	Code for format in which drawdown should be printed.
IDDNUN	Package	Unit number on which an unformatted record containing drawdown should be recorded.
IHDDFL	Package	Flag. = 0, neither head nor drawdown will be printed at this time step. ≠ 0, head and drawdown may be printed at the end of the current time step.
IHEDFM	Package	Code for the format in which head should be printed.
IHEDUN	Package	Unit number on which an unformatted record containing head should be recorded.
IOFLG	Package	DIMENSION (NLAY,4), Flags to control printing and recording of head and drawdown for each layer. (NLAY,1) ≠ 0, heads will be printed. (NLAY,2) ≠ 0, drawdown will be printed. (NLAY,3) ≠ 0, heads will be recorded. (NLAY,4) ≠ 0, drawdown will be recorded.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IPFLG	Package	Flag. = 0 means nothing has been printed this time step. ≠ 0 means something has been printed this time step; therefore, a time summary must be printed.
ISTRT	Package	Flag. ≠ 0, starting heads will be saved so that drawdown can be calculated. = 0, starting heads will not be saved.
ITMUNI	Package	Code for time units for this problem: 0 - undefined 1 - seconds 2 - minutes 3 - hours 4 - days 5 - years

List of Variables for Module BAS10T (Continued)

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
KPER	Global	Stress period counter.
KSTP	Global	Time step counter. Reset at the start of each stress period.
MSUM	Global	Counter for budget entries and labels in VBVL and VBNM.
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.
PERTIM	Package	Elapsed time during the current stress period.
STRT	Package	DIMENSION (NCOL,NROW,NLAY), Starting head.
TOTIM	Package	Elapsed time in the simulation.
VBNM	Global	DIMENSION (4,20), Labels for entries in the volumetric budget.
VBVL	Global	DIMENSION (4,20), Entries for the volumetric budget. For flow component N, the values in VBVL are: (1,N) Rate for the current time step into the flow field. (2,N) Rate for the current time step out of the flow field. (3,N) Volume into the flow field during simulation. (4,N) Volume out of the flow field during simulation.

## Narrative for Module SBAS1D

Module SBAS1D is called by module BAS10T to calculate and write drawdown for every cell in certain layers in the grid. The module is called at the end of each time step if the head and drawdown flag (IHDDFL) is set. It calculates drawdown only if the user has specified that starting heads should be saved.

The layers for which drawdown is to be written are determined by the settings of flags in the table named IOFLG. In IOFLG, there are four flags for each layer. The second flag, if it is set, causes drawdown to be printed. The fourth flag, if it is set, causes drawdown to be recorded.

Module SBAS1D performs its functions in the following order:

1. For each layer, do steps 2-5.
2. If flags indicate that drawdown is not needed for this layer, go on to the next layer.
3. Test flag ISTRT to see if starting heads were saved. Go to either 4 or 5.
4. Starting heads were not saved. Write a message to that effect and STOP.
5. Starting heads were saved. Calculate drawdown for this layer.
6. For each layer, if drawdown is to be printed, call module ULAPRS or ULAPRW, depending on the format requested (IDDNFM), to print drawdown.
7. For each layer, if drawdown is to be recorded, call module ULASAV to write the drawdown to the unit specified in IDDNUN.
8. RETURN.

## Flow Chart for Module SBAS1D


IOFLG is a table containing one entry for each layer. Each element consists of four flags which, when set, cause (1) head to be printed, (2) drawdown to be printed, (3) head to be recorded, and (4) drawdown to be recorded.

ULAPRW is a utility module which prints a value for each cell in a wrap format. In the wrap format, all values for one row are printed before any values for the next row.

ULAPRS is a utility module which prints a value for each cell in the layer in a strip format. In the strip format, all values in a group of N columns are printed before any values in the next N columns are printed.

ULASAV is a utility module which records a value for each cell in a layer.

IDDNUN is a unit number, specified by the user, on which drawdown will be recorded.



```

SUBROUTINE SBAS1D(HNEW, BUFF, IOFLG, KSTP, KPER, NCOL, NROW,
1  NLAY, IOUT, IDDNFM, IDDNUN, STRT, ISTRT, IBOUND, IPFLG,
2  PERTIM, TOTIM)
C-----VERSION 1630 15MAY1987 SBAS1D
C *****
C CALCULATE PRINT AND RECORD DRAWDOWNS
C *****
C
C SPECIFICATIONS
C -----
C CHARACTER*4 TEXT
C DOUBLE PRECISION HNEW
C
C DIMENSION HNEW(NCOL, NROW, NLAY), IOFLG(NLAY, 4), TEXT(4),
1  BUFF(NCOL, NROW, NLAY), STRT(NCOL, NROW, NLAY),
2  IBOUND(NCOL, NROW, NLAY)
C
C DATA TEXT(1), TEXT(2), TEXT(3), TEXT(4) /' ', ' ', ' ', 'DRAW',
1  'DOWN'/
C -----
C
C1-----FOR EACH LAYER CALCULATE DRAWDOWN IF PRINT OR RECORD
C1-----IS REQUESTED
C DO 59 K=1, NLAY
C
C2-----IS DRAWDOWN NEEDED FRO THIS LAYER?
C IF(IOFLG(K, 2).EQ.0 .AND. IOFLG(K, 4).EQ.0) GO TO 59
C
C3-----DRAWDOWN IS NEEDED. WERE STARTING HEADS SAVED?
C IF(ISTRT.NE.0) GO TO 53
C
C4-----STARTING HEADS WERE NOT SAVED. PRINT MESSAGE AND STOP.
C WRITE(IOUT, 52)
C 52 FORMAT(1H0, 'CANNOT CALCULATE DRAWDOWN BECAUSE START',
1  ' HEADS WERE NOT SAVED')
C STOP
C
C5-----CALCULATE DRAWDOWN FOR THE LAYER.
C 53 DO 58 I=1, NROW
C DO 58 J=1, NCOL
C HSING=HNEW(J, I, K)
C BUFF(J, I, K)=HSING
C IF(IBOUND(J, I, K).NE.0) BUFF(J, I, K)=STRT(J, I, K)-HSING
C 58 CONTINUE
C 59 CONTINUE
C
C6-----FOR EACH LAYER: DETERMINE IF DRAWDOWN SHOULD BE PRINTED.
C6-----IF SO THEN CALL ULAPRS OR ULAPRW TO PRINT DRAWDOWN.
C DO 69 K=1, NLAY
C KK=K
C IF(IOFLG(K, 2).EQ.0) GO TO 69
C IF(IDDNFM.LT.0) CALL ULAPRS(BUFF(1, 1, K), TEXT(1), KSTP, KPER,
1  NCOL, NROW, KK, -IDDNFM, IOUT)
C IF(IDDNFM.GE.0) CALL ULAPRW(BUFF(1, 1, K), TEXT(1), KSTP, KPER,
1  NCOL, NROW, KK, IDDNFM, IOUT)
C IPFLG=1
C 69 CONTINUE
C
C7-----FOR EACH LAYER: DETERMINE IF DRAWDOWN SHOULD BE RECORDED.
C7-----IF SO THEN CALL ULASAV TO RECORD DRAWDOWN.
C IFIRST=1
C IF(IDDNUN.LE.0) GO TO 80
C DO 79 K=1, NLAY
C KK=K
C IF(IOFLG(K, 4).LE.0) GO TO 79
C IF(IFIRST.EQ.1) WRITE(IOUT, 74) IDDNUN, KSTP, KPER
C 74 FORMAT(1H0, 'DRAWDOWN WILL BE SAVED ON UNIT', I3,
1  ' AT END OF TIME STEP', I3, ' STRESS PERIOD', I3)
C IFIRST=0
C CALL ULASAV(BUFF(1, 1, K), TEXT(1), KSTP, KPER, PERTIM, TOTIM, NCOL,
1  NROW, KK, IDDNUN)
C 79 CONTINUE
C
C8-----RETURN
C 80 RETURN
C END

```

List of Variables for Module SBAS1D

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
BUFF	Global	DIMENSION (NCOL,NROW,NLAY), Buffer used to accumulate information before printing or recording it.
HNEW	Global	DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration.
HSING	Module	Single-precision temporary field for HNEW (J,I,K).
I	Module	Index for rows.
IBOUND	Global	DIMENSION (NCOL,NROW,NLAY), Status of each cell. < 0, constant-head cell = 0, inactive cell > 0, variable-head cell
IDDNFM	Package	Code for format in which drawdown should be printed.
IDDNUN	Package	Unit number on which an unformatted record containing drawdown should be recorded.
IFIRST	Module	Flag to indicate that a notice should be printed when drawdown is recorded.
IOFLG	Package	DIMENSION (NLAY,4), Flags to control printing and recording of head and drawdown for each layer. (NLAY,1) ≠ 0, heads will be printed. (NLAY,2) ≠ 0, drawdown will be printed. (NLAY,3) ≠ 0, heads will be recorded. (NLAY,4) ≠ 0, drawdown will be recorded.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IPFLG	Package	Flag. = 0 means nothing has been printed this time step. ≠ 0 means something has been printed this time step; therefore, a time summary must be printed.
ISTRT	Package	Flag. ≠ 0, starting heads will be saved so that drawdown can be calculated. = 0, starting heads will not be saved.
J	Module	Index for columns.
K	Module	Index for layers.
KK	Module	Temporary variable set equal to K. KK is used as an actual argument in subroutine calls to avoid using the DO loop variable K as an argument, which causes problems with some compilers.
KPER	Global	Stress period counter.
KSTP	Global	Time step counter. Reset at the start of each stress period.
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.
PERTIM	Package	Elapsed time during the current stress period.
STRT	Package	DIMENSION (NCOL,NROW,NLAY), Starting head.
TEXT	Module	Label to be printed or recorded with array data.
TOTIM	Package	Elapsed time in the simulation.

## Narrative for Module SBAS1H

Module SBAS1H prints and records head for every cell in certain layers in the grid. It is called by module BAS10T at the end of each time step if the head and drawdown flag (IHDDFL) is set. The layers for which head is written is controlled by the settings of flags in the table named IOFLG. In IOFLG, there are four flags for each layer. The first flag, if it is set, causes head for the corresponding layer to be printed. The third flag, if it is set, causes head to be recorded.

Module SBAS1H performs its functions in the following order:

1. For each layer, DO STEPS 2-4.
2. Test the flag table (IOFLG) to see if heads should be printed for this layer. If so, DO STEPS 3 AND 4.
3. Copy heads for this layer (which are contained in the double-precision array (HNEW)) into the single-precision buffer array (BUFF).
4. Depending on the print-format code, call either module ULAPRW or ULAPRS to print the contents of the buffer array.
5. Test the unit number for recording heads (IHEDUN) to see if it is positive. If it is not positive, heads will not be recorded (SKIP STEPS 6-9). If it is positive, heads may be recorded in accordance with the setting of flags in the IOFLG array. DO STEPS 6-9.
6. For each layer, DO STEPS 7-9.
7. If flags in IOFLG indicate that heads are not to be recorded for this layer, move on to the next layer.
8. Copy heads from the HNEW array (double-precision) to the BUFF array (single-precision).
9. Call module ULASAV to record the heads on unit IHEDUN.
10. RETURN.

Flow Chart for Module SBAS1H


IOFLG is a table containing one entry for each layer. Each element consists of four flags which, when set, cause (1) head to be printed, (2) drawdown to be printed, (3) head to be recorded, and (4) drawdown to be recorded.

ULAPRW is a utility module which prints a value for each cell in the layer in a wrap format. In the wrap format, all values for one row are printed before any values for the next row.

ULAPRS is a utility module which prints a value for each cell in the layer in a strip format. In the strip format, all values in a group of N columns are printed before any values in the next N columns are printed.

ULASAV is a utility module which records a value for each cell in a layer.

IHDUN is a unit number, specified by the user, on which heads will be recorded. If IHEDUN is less than or equal to zero, heads will not be recorded.



```

      SUBROUTINE SBAS1H(HNEW,BUFF,IOFLG,KSTP,KPER,NCOL,NROW,
1      NLAY,IOUT,IHEDFM,IHEDUN,IPFLG,PERTIM,TOTIM)
C
C-----VERSION 1653 15MAY1987 SBAS1H
C *****
C PRINT AND RECORD HEADS
C *****
C
C SPECIFICATIONS
C -----
C CHARACTER*4 TEXT
C DOUBLE PRECISION HNEW
C
C DIMENSION HNEW(NCOL,NROW,NLAY),IOFLG(NLAY,4),TEXT(4),
1 BUFF(NCOL,NROW,NLAY)
C
C DATA TEXT(1),TEXT(2),TEXT(3),TEXT(4) /' ',',',',',',
1 'HEAD'/
C -----
C
C1-----FOR EACH LAYER: PRINT HEAD IF REQUESTED.
      DO 39 K=1,NLAY
      KK=K
C
C2-----TEST IOFLG TO SEE IF HEAD SHOULD BE PRINTED.
      IF(IOFLG(K,1).EQ.0) GO TO 39
      IPFLG=1
C
C3-----COPY HEADS FOR THIS LAYER INTO BUFFER.
      DO 32 I=1,NROW
      DO 32 J=1,NCOL
      BUFF(J,I,1)=HNEW(J,I,K)
      32 CONTINUE
C
C4-----CALL UTILITY MODULE TO PRINT CONTENTS OF BUFFER.
      IF(IHEDFM.LT.0) CALL ULAPRS(BUFF,TEXT(1),KSTP,KPER,NCOL,NROW,KK,
1 --IHEDFM,IOUT)
      IF(IHEDFM.GE.0) CALL ULAPRW(BUFF,TEXT(1),KSTP,KPER,NCOL,NROW,KK,
1 IHEDFM,IOUT)
      39 CONTINUE
C
C5-----IF UNIT FOR RECORDING HEADS <= 0: THEN RETURN.
      IF(IHEDUN.LE.0)GO TO 50
      IFIRST=1
C
C6-----FOR EACH LAYER: RECORD HEAD IF REQUESTED.
      DO 49 K=1,NLAY
      KK=K
C
C7-----CHECK IOFLG TO SEE IF HEAD FOR THIS LAYER SHOULD BE RECORDED.
      IF(IOFLG(K,3).LE.0) GO TO 49
      IF(IFIRST.EQ.1) WRITE(IOUT,41) IHEDUN,KSTP,KPER
      41 FORMAT(1H0,'HEAD WILL BE SAVED ON UNIT',I3,' AT END OF TIME STEP',
1 I3,', STRESS PERIOD',I3)
      IFIRST=0
C
C8-----COPY HEADS FOR THIS LAYER INTO BUFFER.
      DO 44 I=1,NROW
      DO 44 J=1,NCOL
      BUFF(J,I,1)=HNEW(J,I,K)
      44 CONTINUE
C
C9-----RECORD CONTENTS OF BUFFER ON UNIT=IHEDUN
      CALL ULASAV(BUFF,TEXT(1),KSTP,KPER,PERTIM,TOTIM,NCOL,NROW,KK,
1 IHEDUN)
      49 CONTINUE
C
C10-----RETURN
      50 RETURN
      END

```

List of Variables for Module SBAS1H

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
BUFF	Global	DIMENSION (NCOL,NROW,NLAY), Buffer used to accumulate information before printing or recording it.
HNEW	Global	DIMENSION (NCOL,NROW,NLAY), Most recent estimate of head in each cell. HNEW changes at each iteration.
I	Module	Index for rows.
IFIRST	Module	Flag which, if set (equal to 1), indicates that a notice should be printed when head is recorded.
IHEDFM	Package	Code for format in which head should be printed.
IHEDUN	Package	Unit number on which an unformatted record containing head should be recorded.
IOFLG	Package	DIMENSION (NLAY,4), Flags to control printing and recording of head and drawdown for each layer. (NLAY,1) ≠ 0, heads will be printed. (NLAY,2) ≠ 0, drawdown will be printed. (NLAY,3) ≠ 0, heads will be recorded. (NLAY,4) ≠ 0, drawdown will be recorded.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
IPFLG	Package	Flag. = 0 means nothing has been printed this time step. ≠ 0 means something has been printed this time step; therefore, a time summary must be printed.
J	Module	Index for columns.
K	Module	Index for layers.
KK	Module	Temporary variable set equal to K. KK is used as an actual argument in subroutine calls to avoid using the DO loop variable K as an argument, which causes problems with some compilers.
KPER	Global	Stress period counter.
KSTP	Global	Time step counter. Reset at the start of each stress period.
NCOL	Global	Number of columns in the grid.
NLAY	Global	Number of layers in the grid.
NROW	Global	Number of rows in the grid.
PERTIM	Package	Elapsed time during the current stress period.
TEXT	Module	Label to be printed or recorded with array data.
TOTIM	Package	Elapsed time in the simulation.

## Narrative for Module SBAS1I

Module SBAS1I initializes the Output Control System. If the user does not opt to control output, the formats for printing head and drawdown are set to the default format and flags are set so that, whenever heads or drawdowns are printed, they are printed for all layers. If the user does opt to control output, the formats for printing and the unit numbers for recording head and drawdown are read.

A table named IOFLG contains one entry for each layer in the grid. Each entry consists of four flags corresponding to four operations: (1) head print, (2) drawdown print, (3) head record, and (4) drawdown record. The module BAS1OT examines the table and, for each layer, performs only the operations for which the corresponding flags are set (equal to one). This module (SBAS1I) sets the head-print flag if the user opts for default output. If starting heads are saved, it also sets the drawdown-print flag. If the user opts to control output, the flags in IOFLG are read at each time step.


Module SBAS1I performs its functions in the following order:

1. Test the unit number for Output Control (IUNIT (12)), which is known in this module by the name INOC, to see if it is positive. If it is positive, the Output Control option is active and output specification will be read from the unit number contained in INOC. If it is not positive, the Output Control option is not active and flags are set to defaults. GO TO 2 OR 3.
2. Output Control is active. Read and print the head-print format code (IHEDFM), the drawdown-print format code (IDDNFM), the unit number to record heads (IHEDUN), and the unit number to record drawdown (IDDNUN). GO TO 6. Note: The formats and associated codes are listed in the Input Instructions for Output Control.
3. Output Control is inactive. Print a message listing the defaults.
4. Set the print-format codes (IHEDFM and IDDNFM) equal to zero to get the default format.
5. Set the flags in IOFLG so that head and drawdown are printed for all layers.
6. RETURN.

## Flow Chart for Module SBAS11

INOC is the input unit for Output Control. It is the same as element 12 in the IUNIT table. When it is greater than zero, Output Control is active--the user will provide output specifications. When it is less than or equal to zero, Output Control is inactive--output will be controlled by default.

IOFLG is a table containing one entry for each layer. Each entry consists of four flags which, when set, cause (1) head to be printed, (2) drawdown to be printed, (3) head to be recorded, and (4) drawdown to be recorded.



```

SUBROUTINE SBAS1I(NLAY, ISTRT, IOFLG, INOC, IOUT, IHEDFM,
1          IDDNFM, IHEDUN, IDDNUN)
C
C-----VERSION 1531 12MAY1987 SBAS1I
C *****
C SET UP OUTPUT CONTROL
C *****
C
C SPECIFICATIONS:
C -----
C DIMENSION IOFLG(NLAY,4)
C -----
C
C1-----TEST UNIT NUMBER FROM IUNIT (INOC) TO SEE IF OUTPUT
C1-----CONTROL IS ACTIVE.
          IF(INOC.LE.0) GO TO 600
C
C2-----READ AND PRINT FORMATS FOR PRINTING AND UNIT NUMBERS FOR
C2-----RECORDING HEADS AND DRAWDOWN. THEN RETURN.
          500 READ (INOC,1)IHEDFM, IDDNFM, IHEDUN, IDDNUN
              1 FORMAT (4I10)
              WRITE (IOUT,3)IHEDFM, IDDNFM
              3 FORMAT (1H0, 'HEAD PRINT FORMAT IS FORMAT NUMBER', I4,
              1 ' DRAWDOWN PRINT FORMAT IS FORMAT NUMBER', I4)
              WRITE (IOUT,4)IHEDUN, IDDNUN
              4 FORMAT (1H0, 'HEADS WILL BE SAVED ON UNIT', I3,
              1 ' DRAWDOWNS WILL BE SAVED ON UNIT', I3)
              WRITE(IOUT,561)
          561 FORMAT(1H0, 'OUTPUT CONTROL IS SPECIFIED EVERY TIME STEP')
              GO TO 1000
C
C3-----OUTPUT CONTROL IS INACTIVE. PRINT A MESSAGE LISTING DEFAULTS.
          600 WRITE(IOUT,641)
          641 FORMAT(1H0, 'DEFAULT OUTPUT CONTROL -- THE FOLLOWING OUTPUT',
          1 ' COMES AT THE END OF EACH STRESS PERIOD:')
              WRITE(IOUT,642)
          642 FORMAT(1X, 'TOTAL VOLUMETRIC BUDGET')
              WRITE(IOUT,643)
          643 FORMAT(1X, 10X, 'HEAD')
              IF(ISTRT.NE.0)WRITE(IOUT,644)
          644 FORMAT(1X, 10X, 'DRAWDOWN')
C
C4-----SET THE FORMAT CODES EQUAL TO THE DEFAULT FORMAT.
          IHEDFM=0
          IDDNFM=0
C
C5-----SET DEFAULT FLAGS IN IOFLG SO THAT HEAD (AND DRAWDOWN) IS
C5-----PRINTED FOR EVERY LAYER.
          ID=0
          IF(ISTRT.NE.0) ID=1
          670 DO 680 K=1, NLAY
              IOFLG(K,1)=1
              IOFLG(K,2)=ID
              IOFLG(K,3)=0
              IOFLG(K,4)=0
          680 CONTINUE
              GO TO 1000
C
C6-----RETURN
1000 RETURN
      END

```

List of Variables for Module SBAS1I

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
ID	Module	Flag to show if STRT was saved (one means yes; zero means no).
IDDNFM	Package	Code for format in which drawdown should be printed.
IDDNUN	Package	Unit number on which an unformatted record containing drawdown should be recorded.
IHEDFM	Package	Code for format in which head should be printed.
IHEDUN	Package	Unit number on which an unformatted record containing head should be recorded.
INOC	Package	Unit number from which input to output control option will be read.
IOFLG	Package	DIMENSION (NLAY,4), Flags to control printing and recording of head and drawdown for each layer. (NLAY,1) ≠ 0, heads will be printed. (NLAY,2) ≠ 0, drawdown will be printed. (NLAY,3) ≠ 0, heads will be recorded. (NLAY,4) ≠ 0, drawdown will be recorded.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
ISTR	Package	Flag. ≠ 0, starting heads will be saved so that drawdown can be calculated. = 0, starting heads will not be saved.
K	Module	Index for layers
NLAY	Global	Number of layers in the grid.

### Narrative for Module SBAS1T

Submodule SBAS1T prints a time summary which consists of the time-step length and the elapsed time in seconds, minutes, hours, days, and years. The program can use any consistent set of time units. However, the user is given the option to specify the time units that he is using and the program converts those units to all other convenient units. The user specifies time units (ITMUNI) in module BAS1DF.

1. Use the time-unit indicator (ITMUNI) to determine the conversion factor (CNV) needed to convert time to seconds.

2. If the conversion factor is equal to zero, nonstandard time units are being used.

- (a) Print the time-step length and the elapsed time in the nonstandard units.

- (b) RETURN.

3. Calculate the length of the time step and the elapsed times in seconds.

4. Calculate the time-step length and the elapsed times in minutes, hours, days, and years.

5. Print the time-step length and the elapsed times in all time units.

6. RETURN.

```

SUBROUTINE SBAS1T(KSTP,KPER,DELT,PERTIM,TOTIM,ITMUNI,IOUT)
C
C
C-----VERSION 0837 09APR1982 SBAS1T
C *****
C PRINT SIMULATION TIME
C *****
C
C SPECIFICATIONS:
C -----
C -----
C WRITE(IOUT,199) KSTP,KPER
199 FORMAT(1H0,///10X,'TIME SUMMARY AT END OF TIME STEP',I3,
1 ' IN STRESS PERIOD',I3)
C
C1-----USE TIME UNIT INDICATOR TO GET FACTOR TO CONVERT TO SECONDS.
CNV=0.
IF(ITMUNI.EQ.1) CNV=1.
IF(ITMUNI.EQ.2) CNV=60.
IF(ITMUNI.EQ.3) CNV=3600.
IF(ITMUNI.EQ.4) CNV=86400.
IF(ITMUNI.EQ.5) CNV=31557600.
C
C2-----IF FACTOR=0 THEN TIME UNITS ARE NON-STANDARD.
IF(CNV.NE.0.) GO TO 100
C
C2A-----PRINT TIMES IN NON-STANDARD TIME UNITS.
WRITE(IOUT,301) DELT,PERTIM,TOTIM
301 FORMAT(21X,' TIME STEP LENGTH =',G15.6/
1 21X,' STRESS PERIOD TIME =',G15.6/
2 21X,'TOTAL SIMULATION TIME =',G15.6)
C
C2B-----RETURN
RETURN
C
C3-----CALCULATE LENGTH OF TIME STEP & ELAPSED TIMES IN SECONDS.
100 DELSEC=CNV*DELT
TOTSEC=CNV*TOTIM
PERSEC=CNV*PERTIM
C
C4-----CALCULATE TIMES IN MINUTES,HOURS,DAYS AND YEARS.
DELMN=DELSEC/60.
DELHR=DELMN/60.
DELDY=DELHR/24.
DELYR=DELDY/365.25
TOTMN=TOTSEC/60.
TOTHR=TOTMN/60.
TOTDY=TOTHR/24.
TOTYR=TOTDY/365.25
PERMN=PERSEC/60.
PERHR=PERMN/60.
PERDY=PERHR/24.
PERYR=PERDY/365.25
C
C5-----PRINT TIME STEP LENGTH AND ELAPSED TIMES IN ALL TIME UNITS.
WRITE(IOUT,200)
200 FORMAT(27X,' SECONDS MINUTES HOURS',10X,
1 'DAYS YEARS'/27X,75('-'))
WRITE(IOUT,201) DELSEC,DELMN,DELHR,DELDY,DELYR
201 FORMAT(1X,' TIME STEP LENGTH',5X,5G15.6)
WRITE(IOUT,202) PERSEC,PERMN,PERHR,PERDY,PERYR
202 FORMAT(1X,' STRESS PERIOD TIME',5X,5G15.6)
WRITE(IOUT,203) TOTSEC,TOTMN,TOTHR,TOTDY,TOTYR
203 FORMAT(1X,'TOTAL SIMULATION TIME',5X,5G15.6)
C
C6-----RETURN
RETURN
END

```

List of Variables for Module SBAS1T

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
CNV	Module	Factor to convert elapsed time from units, specified by the user, to seconds.
DELDY	Module	Length of the time step in days.
DELHR	Module	Length of the time step in hours.
DELMN	Module	Length of the time step in minutes.
DELSEC	Module	Length of the time step in seconds.
DELT	Global	Length of the current time step.
DELYR	Module	Length of the time step in years.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
ITMUNI	Package	Code for time units for this problem: 0 - undefined 1 - seconds 2 - minutes 3 - hours 4 - days 5 - years
KPER	Global	Stress period counter.
KSTP	Global	Time step counter. Reset at the start of each stress period.
PERDY	Module	Elapsed time in the stress period in days.
PERHR	Module	Elapsed time in the stress period in hours.
PERMN	Module	Elapsed time in the stress period in minutes.
PERSEC	Module	Elapsed time in the stress period in seconds.
PERTIM	Package	Elapsed time during the current stress period.
PERYR	Module	Elapsed time in the stress period in years.
TOTDY	Module	Elapsed time in the simulation in days.
TOTHR	Module	Elapsed time in the simulation in hours.
TOTIM	Package	Elapsed time in the simulation.
TOTMN	Module	Elapsed time in the simulation in minutes.
TOTSEC	Module	Elapsed time in the simulation in seconds.
TOTYR	Module	Elapsed time in the simulation in years.

## Narrative for Module SBAS1V

Module SBAS1V calculates and prints the overall volumetric budget. The individual entries for the budget, which are calculated by the budget modules in each of the component-of-flow packages, are passed to this module in a table named VBVL.


Each entry in VBVL corresponds to a component-of-flow. It consists of four values: rate of inflow for the current time step, rate of outflow for the current time step, accumulated volume of inflow since the beginning of the simulation, and accumulated volume of outflow since the beginning of the simulation. In this module, the total of all inflow rates (TOTRIN), outflow rates (TOTROT), inflow-accumulated volumes (TOTVIN), and outflow-accumulated volumes (TOTVOT) are calculated. The percent differences between those totals are also calculated and printed. The labels for the entries are supplied by the budget modules in the component-of-flow packages and passed in the table VBNM.

Module SBAS1V performs its functions in the following order:

1. Use the counter MSUM to determine the number of individual budget terms (MSUM1).
2. Clear the four accumulators for rates and volumes. The accumulators are total rate into the system (TOTRIN), total rate out of the system (TOTROT), accumulated volume into the system (TOTVIN), and accumulated volume out of the system (TOTVOT).
3. For each source or sink, add the budget entries (rates and volumes), calculated by the budget modules, to the accumulators.
4. Print the number of the time step and stress period.
5. Print the individual input rates and volumes and their totals.
6. Print the individual output rates and volumes and their totals.
7. Calculate the difference between flow into and out of the simulated-flow system. Calculate the percent difference between input and output rates ( $100 * (TOTRIN - TOTROT) / ((TOTRIN + TOTROT) / 2)$ ). Calculate the percent difference between input and output accumulated volumes ( $100 * (TOTVIN - TOTVOT) / ((TOTVIN + TOTVOT) / 2)$ ).
8. Print the differences and percent differences between input and output rates and volumes.
9. RETURN.

Flow Chart for Module SBAS1V

The inflow and outflow rates for the current time step and accumulated volumes since the beginning of the simulation for each budget entry (component-of-flow) are added to the four accumulators to obtain the total inflow and outflow rates for the current time step and the total accumulated volume of flow in and volume of flow out since the start of the simulation.



```

SUBROUTINE SBAS1V(MSUM, VBNM, VBVL, KSTP, KPER, IOUT)
C
C
C-----VERSION 1531 12MAY1987 SBAS1V
C*****
C PRINT VOLUMETRIC BUDGET
C*****
C
C SPECIFICATIONS:
C-----
C CHARACTER*4 VBNM
C DIMENSION VBNM(4,20),VBVL(4,20)
C-----
C
C1-----DETERMINE NUMBER OF INDIVIDUAL BUDGET ENTRIES.
      MSUM1=MSUM-1
      IF(MSUM1.LE.0) RETURN
C
C2-----CLEAR RATE AND VOLUME ACCUMULATORS.
      TOTRIN=0.
      TOTROT=0.
      TOTVIN=0.
      TOTVOT=0.
C
C3-----ADD RATES AND VOLUMES (IN AND OUT) TO ACCUMULATORS.
      DO 100 L=1,MSUM1
        TOTRIN=TOTRIN+VBVL(3,L)
        TOTROT=TOTROT+VBVL(4,L)
        TOTVIN=TOTVIN+VBVL(1,L)
        TOTVOT=TOTVOT+VBVL(2,L)
      100 CONTINUE
C
C4-----PRINT TIME STEP NUMBER AND STRESS PERIOD NUMBER.
      WRITE(IOUT,260) KSTP,KPER
      WRITE(IOUT,265)
C
C5-----PRINT INDIVIDUAL INFLOW RATES AND VOLUMES AND THEIR TOTALS.
      DO 200 L=1,MSUM1
        WRITE(IOUT,275) (VBNM(I,L),I=1,4),VBVL(1,L),(VBNM(I,L),I=1,4)
          1,VBVL(3,L)
      200 CONTINUE
      WRITE(IOUT,286) TOTVIN,TOTRIN
C
C6-----PRINT INDIVIDUAL OUTFLOW RATES AND VOLUMES AND THEIR TOTALS.
      WRITE(IOUT,287)
      DO 250 L=1,MSUM1

```

```

        WRITE(IOUT,275) (VBNM(I,L),I=1,4),VBVL(2,L),(VBNM(I,L),I=1,4)
        1,VBVL(4,L)
250 CONTINUE
        WRITE(IOUT,298) TOTVOT,TOTROT
C
C7-----CALCULATE THE DIFFERENCE BETWEEN INFLOW AND OUTFLOW.
C
C7A-----CALCULATE DIFFERENCE BETWEEN RATE IN AND RATE OUT.
        DIFFR=TOTRIN-TOTROT
C
C7B-----CALCULATE PERCENT DIFFERENCE BETWEEN RATE IN AND RATE OUT.
        PDIFFR=100.*DIFFR/((TOTRIN+TOTROT)/2)
C
C7C-----CALCULATE DIFFERENCE BETWEEN VOLUME IN AND VOLUME OUT.
        DIFFV=TOTVIN-TOTVOT
C
C7D-----GET PERCENT DIFFERENCE BETWEEN VOLUME IN AND VOLUME OUT.
        PDIFFV=100.*DIFFV/((TOTVIN+TOTVOT)/2)
C
C8-----PRINT DIFFERENCES AND PERCENT DIFFERENCES BETWEEN INPUT
C8-----AND OUTPUT RATES AND VOLUMES.
        WRITE(IOUT,299) DIFFV,DIFFR
        WRITE(IOUT,300) PDIFFV,PDIFFR
C
C9-----RETURN
        RETURN
C
C    ---FORMATS
C
260 FORMAT(1H0,///30X,'VOLUMETRIC BUDGET FOR ENTIRE MODEL AT END OF'
1,' TIME STEP',I3,' IN STRESS PERIOD',I3/30X,77('-'))
265 FORMAT(1H0,19X,'CUMULATIVE VOLUMES',6X,'L**3',37X
1,' RATES FOR THIS TIME STEP',6X,'L**3/T'/20X,18('-'),47X,24('-')
2//26X,' IN:',68X,' IN: '/26X,' ---',68X,' ---')
275 FORMAT(1X,18X,4A4,' =',G14.5,39X,4A4,' =',G14.5)
286 FORMAT(1H0,26X,'TOTAL IN =',G14.5,47X,'TOTAL IN ='
1,G14.5)
287 FORMAT(1H0,24X,'OUT:',67X,'OUT: '/25X,4('-'),67X,4('-'))
298 FORMAT(1H0,25X,'TOTAL OUT =',G14.5,46X,'TOTAL OUT ='
1,G14.5)
299 FORMAT(1H0,26X,'IN - OUT =',G14.5,47X,'IN - OUT =',G14.5)
300 FORMAT(1H0,15X,'PERCENT DISCREPANCY =',F20.2
1,30X,'PERCENT DISCREPANCY =',F20.2,///)
C
        END

```

List of Variables for Module SBAS1V

<u>Variable</u>	<u>Range</u>	<u>Definition</u>
DIFFR	Module	Sum of all inflow rates minus sum of all outflow rates (TOTRIN-TOTROT).
DIFFV	Module	Sum of all inflow volumes minus sum of all outflow volumes (TOTVIN-TOTVOT).
I	Module	Index.
IOUT	Global	Primary unit number for all printed output. IOUT = 6.
KPER	Global	Stress period counter.
KSTP	Global	Time step counter. Reset at the start of each stress period.
L	Module	Index for individual flows.
MSUM	Global	Counter for the budget entries and labels in VBVL and VBNM.
MSUM1	Module	MSUM-1.
PDIFFR	Module	Percent difference between the rate in and rate out.
PDIFFV	Module	Percent difference between the volume in and volume out.
TOTRIN	Module	Accumulator for the total of all inflow rates.
TOTROT	Module	Accumulator for the total of all outflow rates.
TOTVIN	Module	Accumulator for the total of all inflow volumes.
TOTVOT	Module	Accumulator for the total of all outflow volumes.
VBNM	Global	DIMENSION (4,20), Labels for entries in the volumetric budget.
VBVL	Global	DIMENSION (4,20), Entries for the volumetric budget. For flow component N, the values in VBVL are: (1,N) Rate for the current time step into the flow field. (2,N) Rate for the current time step out of the flow field. (3,N) Volume into the flow field during simulation. (4,N) Volume out of the flow field during simulation.